RILs by selfing

Two points

The following is the joint distribution for two points in two-way RILs.

```
Off[General::spell]
```

```
pAA2w = (1 / 2) / (1 + 2 r);
pAB2w = r / (1 + 2 r);
```

To get to the four-way case, we use the following equations.

```
pAA4w = Simplify[ ((1 - r) / 2) pAA2w ]
```

$$\frac{1-r}{4+8\,r}$$

```
pAB4w = Simplify[ (r / 2) pAA2w ]
```

$$\frac{r}{4+8\,r}$$

```
pAC4w = Simplify[ (1 / 4) pAB2w ]
```

$$\frac{r}{4+8\,r}$$

To get to the eight-way case,we use the following equations.

```
pAA8w = Simplify[ ((1 - r) / 2) pAA4w ]
```

$$\frac{(-1+r)^2}{8+16\,r}$$

```
pAB8w = Simplify[ (r / 2) pAA4w ]
```

$$-\frac{(-1+r)\,r}{8+16\,r}$$

```
pAC8w = Simplify[ (1 / 4) pAB4w ]
```

$$\frac{r}{16+32\,r}$$

```
pAE8w = Simplify[ (1 / 4) pAC4w ]
```

$$\frac{r}{16+32\,r}$$

Here is the "recombination fraction" in 2-way RILs.

```
R2w = Simplify[2 pAB2w]
```

$$\frac{2\,r}{1+2\,r}$$

Here is the "recombination fraction" in 4-way RILs

```
R4w = Simplify[ 4 (pAB4w + 2 pAC4w) ]
```

$$\frac{3\,r}{1+2\,r}$$

Here is the "recombination fraction" in 8-way RILs.

```
R8w = Simplify[8 (pAB8w + 2 pAC8w + 4 pAE8w)]
```

$$-\frac{(-4+r)\,r}{1+2\,r}$$

Three points

```
r13 = 2 r (1 - c r);
```

First we consider 2-way RILs, and first we get the expression in terms of the recombination fraction at meiosis.

```
fAA2w[r_] := 1 / 2 / (1 + 2 r);
fAB2w[r_] := r / (1 + 2 r);
```

```
eqns = {pAAA2w + pAAB2w == fAA2w[r],
   pAAB2w + pABA2w == fAB2w[r], pAAA2w + pABA2w == fAA2w[r13]};
```

```
Clear[pAAA2w, pAAB2w, pABA2w];
Solve[eqns, {pAAA2w, pAAB2w, pABA2w}];
```

```
p3pt2w = {pAAA2w, pAAB2w, pABA2w} /. %[[1]];
p3pt2w = Together[p3pt2w]
```

$$\left\{\frac{-1-2\,r+4\,r^2+2\,c\,r^2-4\,c\,r^3}{2\,(1+2\,r)\,(-1-4\,r+4\,c\,r^2)},\ \frac{-r+c\,r^2}{-1-4\,r+4\,c\,r^2},\ \frac{-2\,r^2-c\,r^2+2\,c\,r^3}{(1+2\,r)\,(-1-4\,r+4\,c\,r^2)}\right\}$$

```
R = R2w;
Simplify[2 p3pt2w[[3]] / R^2]
```

$$\frac{2+c+4\,r-4\,c\,r^2}{2+8\,r-8\,c\,r^2}$$

Now let's get the expression in terms of the recombination fraction in the RIL.

```
Clear[R];
r = R / (2 - 2 R);
co2w = Simplify[2 p3pt2w[[3]] / R^2]
```

$$-\frac{2+c-2\,R-2\,c\,R}{2\,(-1+(1+c)\,R^2)}$$

We now turn to the case of four-way RILs.

```
Clear[r];
pAAA4w = Simplify[p3pt2w[[1]] (1 - 2 r + c r^2) / 2];
pAAB4w = Simplify[p3pt2w[[1]] r (1 - c r) / 2];
pAAC4w = Simplify[p3pt2w[[2]] (1 - r) / 4];
pABA4w = Simplify[p3pt2w[[1]] c r^2 / 2];
pABC4w = Simplify[p3pt2w[[2]] r / 4];
pACB4w = Simplify[p3pt2w[[3]] r (1 - c r) / 2];
pACA4w = Simplify[p3pt2w[[3]] (1 - 2 r (1 - c r)) / 4];
```

```
p4pt2w =
 {pAAA4w, pAAB4w, pAAC4w, pABA4w, pABC4w, pACB4w, pACA4w}
```

$$\Big\{-\frac{(1-2r+cr^2)(1+2r-2(2+c)r^2+4cr^3)}{4(1+2r)(-1-4r+4cr^2)},$$
$$\frac{r(-1+cr)(1+2r-2(2+c)r^2+4cr^3)}{4(1+2r)(-1-4r+4cr^2)},$$
$$-\frac{(-1+r)r(-1+cr)}{4(-1-4r+4cr^2)}, -\frac{cr^2(1+2r-2(2+c)r^2+4cr^3)}{4(1+2r)(-1-4r+4cr^2)},$$
$$\frac{r^2-cr^3}{4+16r-16cr^2}, -\frac{r^3(-1+cr)(-2+c(-1+2r))}{2(1+2r)(-1-4r+4cr^2)},$$
$$\frac{r^2(1-2r+2cr^2)(-2+c(-1+2r))}{4(1+2r)(-1-4r+4cr^2)}\Big\}$$

Now we can calculate the coincidence for 4-way RILs by selfing.

```
Clear[r, R];
R = R4w;
co4w = Simplify[(4 pABA4w + 16 pABC4w + 8 pACB4w + 8 pACA4w) / R^2]
```

$$-\frac{(1+2r)(8(1+r)+2c^2r^2(-1+2r)-3c(-1+2r+4r^2))}{9(-1-4r+4cr^2)}$$

And then we can re-expression that in terms of R.

```
Clear[R];
r = R / (3 - 2 R);
Simplify[co4w]
```

$$\frac{8(3-2R)^2(-3+R)+2c^2(3-4R)R^2+3c(-27+72R-48R^2+8R^3)}{3(3-2R)^2(-9+4(1+c)R^2)}$$

```
Simplify[co4w /. {R → 0}]
```

$\frac{1}{9}\ (8 + 3\ c)$

Now we move on to the case of 8-way RILs by selfing. First the three-point probabilities.

```
Clear[r];
pAAA8w = Simplify[pAAA4w (1 / 2) (1 - 2 r + c r^2)];
pAAB8w = Simplify[pAAA4w (1 / 2) r (1 - c r)];
pABA8w = Simplify[pAAA4w (1 / 2) c r^2];
pAAC8w = Simplify[pAAB4w (1 / 4) (1 - r)];
pACA8w = Simplify[pABA4w (1 / 4) (1 - 2 r (1 - c r))];
pAAE8w = Simplify[pAAC4w (1 / 4) (1 - r)];
pAEA8w = Simplify[pACA4w (1 / 4) (1 - 2 r (1 - c r ))];
pABC8w = Simplify[pAAB4w (1 / 4) r];
pACB8w = Simplify[pABA4w (1 / 2) r (1 - c r)];
pABE8w = Simplify[pAAC4w (1 / 4) r];
pAEB8w = Simplify[pACA4w (1 / 2) r (1 - c r)];
pACE8w = Simplify[pABC4w / 8];
pAEC8w = Simplify[pACB4w / 8];
```

```
p3pt8w =
 FullSimplify[{pAAA8w, pAAB8w, pABA8w, pAAC8w, pACA8w, pAAE8w,
   pAEA8w, pABC8w, pACB8w, pABE8w, pAEB8w, pACE8w, pAEC8w}]
```

$$\left\{-\frac{(1+r(-2+c\,r))^2\,(1+2r(-1+2r)(-1+c\,r))}{8(1+2r)(-1+4r(-1+c\,r))},\right.$$
$$\frac{r(-1+c\,r)(1+r(-2+c\,r))(1+2r(-1+2r)(-1+c\,r))}{8(1+2r)(-1+4r(-1+c\,r))},$$
$$-\frac{c\,r^2\,(1+r(-2+c\,r))(1+2r(-1+2r)(-1+c\,r))}{8(1+2r)(-1+4r(-1+c\,r))},$$
$$-\frac{(-1+r)\,r\,(-1+c\,r)(1+2r(-1+2r)(-1+c\,r))}{16(1+2r)(-1+4r(-1+c\,r))},$$
$$-\frac{c\,r^2\,(1+2r(-1+c\,r))(1+2r(-1+2r)(-1+c\,r))}{16(1+2r)(-1+4r(-1+c\,r))},$$
$$\frac{(-1+r)^2\,r\,(-1+c\,r)}{16(-1+4r(-1+c\,r))},\ \frac{r^2\,(-2+c(-1+2r))(1+2r(-1+c\,r))^2}{16(1+2r)(-1+4r(-1+c\,r))},$$
$$\frac{r^2\,(-1+c\,r)(1+2r(-1+2r)(-1+c\,r))}{16(1+2r)(-1+4r(-1+c\,r))},$$
$$\frac{c\,r^3\,(-1+c\,r)(1+2r(-1+2r)(-1+c\,r))}{8(1+2r)(-1+4r(-1+c\,r))},\ -\frac{(-1+r)\,r^2\,(-1+c\,r)}{16(-1+4r(-1+c\,r))},$$
$$-\frac{r^3\,(-1+c\,r)(-2+c(-1+2r))(1+2r(-1+c\,r))}{8(1+2r)(-1+4r(-1+c\,r))},$$
$$\left.\frac{r^2\,(-1+c\,r)}{32(-1+4r(-1+c\,r))},\ -\frac{r^3\,(-1+c\,r)(-2+c(-1+2r))}{16(1+2r)(-1+4r(-1+c\,r))}\right\}$$

Now, the three-point coincidence for 8-way RILs by selfing.

```
R = R8w;
co8w = FullSimplify[
  (1 - 8 pAAA8w - 16 pAAB8w - 32 pAAC8w - 64 pAAE8w) / R^2]
```

$$-\big((1+2r)\,(2c^3r^4(-1+2r)+c^2r^2(-3+2(9-10r)r)+2(7-8(-1+r)r)+4c(1+r(-3+8(-1+r)r)))\big)\,/\,\big((-4+r)^2\,(-1+4r(-1+c\,r))\big)$$

Expressing this in terms of R is exceedingly ugly.

```
Clear[R];
r = (2 - R) - Sqrt[(2 - R)^2 - R];
FullSimplify[co8w]
```

$$-\Big(\Big(5-2\sqrt{(-4+R)\ (-1+R)}-2R\Big)\Big(-82+48\sqrt{(-4+R)\ (-1+R)}-32\Big(-4+\sqrt{(-4+R)\ (-1+R)}\Big)R-32R^2-2c^3\Big(-2+\sqrt{(-4+R)\ (-1+R)}+R\Big)^4\Big(-3+2\sqrt{(-4+R)\ (-1+R)}+2R\Big)-c^2\Big(-2+\sqrt{(-4+R)\ (-1+R)}+R\Big)^2\Big(127-62\sqrt{(-4+R)\ (-1+R)}+2R\Big(-81+20\sqrt{(-4+R)\ (-1+R)}+20R\Big)\Big)+4c\Big(1-\Big(-2+\sqrt{(-4+R)\ (-1+R)}+R\Big)\Big(45-24\sqrt{(-4+R)\ (-1+R)}+16R\Big(-4+\sqrt{(-4+R)\ (-1+R)}+R\Big)\Big)\Big)\Big)\Big)\Big/\Big(\Big(2+\sqrt{(-4+R)\ (-1+R)}+R\Big)^2\Big(-1+4\Big(-2+\sqrt{(-4+R)\ (-1+R)}+R\Big)\Big(1+c\Big(-2+R+\sqrt{4-5R+R^2}\Big)\Big)\Big)\Big)$$